Gongemeisteren

Skrevet av: Geir Arne Hjelle

Oversatt av: Gro Anette Vestre

Kurs: Scratch


Tema: Blokkbasert, Spill

Fag: Matematikk, Programmering

Klassetrinn: 1.-4. klasse, 5.-7. klasse, 8.-10. klasse

Introduksjon

I dag skal me laga eit nyttig spel, nemleg eit spel som vil hjelpa oss å læra andre ting. Me skal få hjelp til å læra gongetabellen!



Steg 1: Læremeisteren

Me skal etterkvart laga eit spel der me får tilfeldige spørsmål frå gongetabellen. Etter at me svarar får me vita om me klarte å svara rett, og me vil også få litt hjelp frå teikningar på skjermen. Men først introduserer me Læremeister som skal hjelpa oss med gongetabellen.

Sjekkliste

Start eit nytt prosjekt og slett kattefiguren.
Legg til ein ny figur ved å klikke på 🍖. Velg ein av <i>Wizard</i> -figurane mest heilt nederst i Menneske -kategorien. Kall figuren Læreimester.
Me skal nå bruka tilfeldig tal-klossar slik at Læremeister kan spørje oss om tilfeldige gongestykker. Skriv dette skriptet:
når grønt flagg klikkes si (tilfeldig tall fra (2) til (10)) i (2) sekunder si [gonger] i (2) sekunder

si (tilfeldig tall fra (2) til (10)) i (2) sekunder

Test prosjektet

Spør Læreimester deg om eit gongestykke?

Klikk på det grøne flaget.

Me skal etterkvart sjå korleis me kan få Læreimesteren til å sei heile teksten, i staden for berre eit ord om gongen.
Foreløpig kan du ikkje svara Læremeister (iallfall ikkje med datamaskinen, prøv

Steg 2: Datamaskinen er ein kalkulator

Du har kanskje ikkje tenkt på det, men datamaskinen er ein veldig flink kalkulator. Det engelske ordet *computer* betyr til og med *reknemaskin*. Me skal nå sjå på korleis me får Scratch til å gonge saman tal.

Sjekkliste

For å få Scratch til å rekne brukar me Operatorer -klossar. Prøv for eksempel å dra gongeklossen - med * -teiknet - til skriptområdet på høyresida. Skriv inn to tal og klikk på klossen. Scratch reknar då ut svaret på gongestykket.



- Me vil nå kombinera gongeklossen med tilfeldig tal-klossen, men for å få dette til treng me ein måte å hugse dei tilfeldige tala: **Variabler**. Gå til Data-kategorien og lag tre nye variabler: tal1, tal2, og riktig svar. La dei gjelde for alle figurar.
- Skriv nå eit **heilt nytt skript** (la det ligge ved sida av det skriptet du skreiv i forrige steg).

```
når jeg mottar [Nytt spørsmål v]
sett [tal1 v] til (tilfeldig tall fra (2) til (10))
sett [tal2 v] til (tilfeldig tall fra (2) til (10))
sett [rett svar v] til ((tal1) * (tal2))
```

Prøv å klikka på skriptet for å testa det (sidan det ikkje startar med eit grønt flag kan me ikkje testa det på den vanlige måten). Om du ser på variablane på scenen skal dei endre seg kvar gong du klikkar på skriptet. Er rett svar rett?



Steg 3: Eit skikkeleg spørsmål

La oss sjå om me kan setje saman desse tala til eit skikkeleg spørsmål.



- Lag ein ny variabel, spørsmål. Også denne skal gjelde for alle figurar.
- Klossen sett sammen kan brukast for å setje saman fleire tal og ord. Me skal nå bruka to sett sammen -klossar på denne måten:

```
sett sammen (sett sammen [] []) []
```

Dette gir oss plass til tre tal eller ord. Her kan me putta inn tal1, teksten gonger og tal2. Pass på at du har mellomrom før og etter gonger, det ser best ut da. Om du klikkar på den første sett sammen -klossen vil du sjå korleis den ferdige teksten blir.



Legg denne klossen til nedst i Nytt spørsmål -skriptet:

```
når jeg mottar [Nytt spørsmål v]
sett [tal1 v] til (tilfeldig tall fra (2) til (10))
sett [tal2 v] til (tilfeldig tall fra (2) til (10)
sett [rett svar v] til ((tal1) * (tal2))
sett [spørsmål v] til (sett sammen (sett sammen (tal1) [ gonger ]) (
```

Nå skal me få Læremeisteren til å stilla oss spørsmålet me har sett saman. Bytt ut det første skriptet (med det grøne flaget) du skreiv med dette:

```
når grønt flagg klikkes
send melding [Nytt spørsmål v] og vent
spør (spørsmål) og vent
```

Test prosjektet

Klikk på det grøne flaget. Stiller Læremeister deg eit skikkeleg spørsmål, for eksempel 9 gonger 6? Er spørsmåla forskjellige kvar gong? Om du vil ha ein liten utfordring kan du prøva å bruka fleire sett sammen -klossar

Steg 4: Er svaret rett då?

Nå som me kan svara på spørsmål vil me og vite om me svarar rett.

slik at spørsmålet blir for eksempel Kva er 9 gonger 6?.

Sjekkliste

- Du ser kanskje at rett svar vises på scenen? Då blir det jo ikkje særleg vanskeleg! Ta bort alle variablane frå scenen ved å gå til Data -kategorien og fjern haka framfor kvar variabel.
- Nå skal me bruka ein hvis ellers -kloss for å gjera forskjellige ting ettersom du svarar rett eller feil på gongestykka. **Utvid** det eine skriptet ditt på denne måten:

```
når grønt flagg klikkes
send melding [Nytt spørsmål v] og vent
spør (spørsmål) og vent
hvis ((svar) = (riktig svar))
    si [Ja, så flink du er!] i (2) sekunder
ellers
    si [Nei, det ble visst feil.] i (2) sekunder
slutt
```

Klossen svar huskar svaret du skriv når Læremeister spør om gongestykket.



Klikk på det grøne flaget. Kva skjer om du svarer rett? Klikk det grøne flaget igjen for å få ei ny oppgåve. Kva skjer om du ikkje svarar rett?

Steg 5: Fleire gongestykke

I staden for å måtte trykke det grøne flaget heile tida, kan me be Læremeister om å stille oss fleire spørsmål!

Sjekkliste

Me brukar først ein gjenta -kloss slik at me kan få fleire oppgåver. Legg merke til at me og sender ein Nytt spørsmål -melding dersom svaret er rett. Dersom svaret er feil spør me det same spørsmålet ein gang til.

```
når grønt flagg klikkes
send melding [Nytt spørsmål v] og vent
gjenta (10) ganger
    spør (spørsmål) og vent
    hvis ((svar) = (riktig svar))
        si [Ja, så flink du er!] i (2) sekunder
        send melding [Nytt spørsmål v] og vent
    ellers
        si [Nei, det ble visst feil.] i (2) sekunder
    slutt
```


- Me kan også telje poeng kvar gong du svarar rett. For å gjera dette treng me ein ny variabel, Poeng. Denne skal gjelde for alle figurar, og denne lar me vera på scenen slik at me ser den.
- Legg til ein kloss i skriptet som set Poeng til 0 rett etter at det grøne flaget vert klikka på.

Klikk på det grøne flaget.

- Får du eit poeng kvar gong du svarar rett?
- Klarar du 10 poeng?

Steg 6: Litt hjelp kanskje ...

Til slutt skal me sjå på korleis Læremeister kan gje oss litt hjelp med gongestykka. Ein måte å tenkje på gongestykker er at me har mange ting som me plasserar i eit rutenett. For eksempel kan me tenkje på 3 gonger 4 som 3 rader med 4 elefanter i kvar som dette:



Dersom me ikkje hugsar kor mykje 3 ganger 4 er, kan me telje elefantar og finne ut at svaret er 12.

Sjekkliste

Legg til ein ny figur som me kan få litt hjelp frå. Du kan velje kva figur du vil, men me har brukt Dyr/Elephant. Gi figuren navnet Hjelpar.

For å teikne eit rutenett med Hjelpere bruker me to gjenta -klossar i tillegg til stemple avtrykk som teiknar Hjelparane på skjermen. Skriv dette skriptet på Hjelper -figuren:

```
når jeg mottar [Tegn hjelper v]
slett
sett størrelse til (20)%
vis
sett y til (140)
gjenta (tal1) ganger
        sett x til (-140)
        gjenta (tal2) ganger
            stemple avtrykk
            endre x med (40)
        slutt
        endre y med (-25)
slutt
skjul
```

Om du har brukt ein annan figur som Hjelpar kan det henda du må bruka nokre andre tal i dette skriptet. Prøv i så fall først å forandra på sett størrelse til 20% -klossen.


Nå skal me teikna dette rutenettet kvar gong me lagar eit nytt spørsmål. Klikk på Læremeister, og legg til ein kloss nedst i Nytt spørsmål-skriptet:

```
når jeg mottar [Nytt spørsmål v]
sett [tal1 v] til (tilfeldig tall fra (2) til (10))
sett [tal2 v] til (tilfeldig tall fra (2) til (10))
sett [rett svar v] til ((tall1) * (tall2))
sett [spørsmål v] til (sett sammen (sett sammen (tal1) [ gonger ]) (
send melding [Tegn hjelper v]
```

Test prosjektet

Klikk på det grøne flaget.

Vert det teikna eit rutenett av hjelparane til kvar oppgåve?



Lisens: CC BY-SA 4.0 (http://creativecommons.org/licenses/by-sa/4.0/deed)